
1 Este tutorial es una versión muy preliminar, por lo que puede contener errores de ortografía o 

mayor. Si encuentra alguno favor acusar a ssouyris@ing.uchile.cl 



MINI TUTORIAL AMPL 
 
El fin de este mini tutorial es dar una escueta introducción AMPL, explicación de 

instalación del software, modelación y obtención de resultados. Este se apoya en 

un ejemplo que se describe posteriormente.2 

 

AMPL es un lenguaje de modelación para programación matemática. 

Este permite llevar de manera relativamente directa problemas de programación 

matemática al computador para que poder resolverlos usando algún paquete de 

optimización o solver. (CPLEX, MINOS, OSL, etc) 

 

¿Cómo instalarlo? 
 

Bajar de http://www.ampl.com/ el archivo amplcml.zip. Seguir las instrucciones de 

instalación. 

 

¿Como ocuparlo? 
 

1. Escribir el archivo que contiene el modelo en cualquier procesador de texto,  

ejemplo.mod. 

2. Escribir el archivo que contiene los datos del modelo, ejemplo.dat. 

3. Guardar ambos archivos en la carpeta amplcml. 

4. Cargar Ampl: ejecutar el archivo sw.exe que esta dentro de la carpeta 

amplcml. Se abre una ventana donde se pueden cargar los archivos de 

texto como entrada, cargar diferentes opciones y resolver. 

5. Cargar el modelo: model ejemplo.mod; 

5.   Cargar datos:  data ejemplo.dat; 

6.   Resolver:  solve; 

7.  Ver resultados display x; 

                                            
2 Para más información visitar http://www.ampl.com/ 



…. 

 

 
Ejemplo sencillo de resolución de un PL usando AMPL 
 

Problema: 

• Empresa fabrica puertas y ventanas utilizando tres recursos: Madera, 
Máquina, Barnizado. 

 

 

 

 

 

 

 

 

 

• Disponibilidad de recursos: 

o Cantidad de madera: 800 m 

o Horas de sierra : 400 

o Horas hombre para barnizado: 300 



• Demanda  

o 200 puertas simples 

o 120 ventanas dobles 

 

• Precios y Costos 

 
Modelo matemático: Se plantea cada parte del modelo seguido de su código en 
AMPL. 
 
Los archivos con el modelo y con los datos son: 
ejemplo.mod y ejemplo.dat 
  
 

• Conjuntos: 
 

o PROD: conjunto de productos 
o REC: conjunto de recursos 

 
*************** 
#conjunto de productos 
set PROD;    
#conjunto de recursos 
set REC;  
****************   

 
• Parámetros: 
 

o crec(j): cantidad disponible del recurso j 
o uso(i,j): cantidad unitaria necesaria a utilizar del recurso j para hacer 

producto i. 
o dda(i): demanda a satisfacer del producto i 
o costo(i): costo unitario de producción del producto i 
o precio(i): precio de venta del producto i 

****************** 
#cantidad de recuso disponible 
param crec {REC} >= 0;  
       
#cantidad necesaria de recurso para cada producto 
param uso {PROD, REC} >= 0; 
       
#demanda del producto 
param dda {PROD} >= 0; 
 
#costo de produccion 
param costo {PROD} >= 0;  
 
#precio de venta del producto   
param precio {PROD} >= 0; 
 
#beneficio neto por unidad producida; 
param ben{p in PROD}:= precio[p]-costo[p]; 
 
************************ 
 
Variables de decisión: 

 
o x(i): cantidad del producto i a producir 

 
************* 
#cuanto produzco  
var x {PROD} >= 0; 
 
*************** 
 
 

• Función Objetivo: Maximizar Utilidades 
 

 
************** 
#maximizar utilidades 
maximize utilidades: sum {p in PROD} (x[p] * ben[p]); 

 
***************** 

 
• Restringido a: 

 
o Satisfacer dda 

o Disponibilidad de recursos 

************************* 
#satisfacer demanda 
subject to satdda {p in PROD}: 
   
 x[p] >=  dda[p]; 
       
#restriccion de recursos 
subject to recmax{r in REC}: 
 
     sum {p in PROD} (x[p]*uso[p,r]) <= crec[r]; 
 
***************************** 
 
Archivo de datos ejemplo.dat: 
 
set PROD := pd ps vd vs; 
 
set REC :=  
mad 
maq 
hhb; 
 
param: crec:= 
mad 860 
maq 440 
hhb 420; 
 
 
param uso: mad maq hhb:= 
 pd 4 1.5  2.0  
 ps 2.5 1  1.2  
 vd 3 2    1.5  
 vs 1.8  0.8  0.8; 
 
param: dda precio costo:= 
pd 0 120 80 
ps 200 80 50 
vd 120 100 75 
vs 0 60 30; 
 
Más ejemplos y tutorial detallado en http://www.ampl.com/ 


